

# An Isabelle/HOL Formalisation of Green’s Theorem

Mohammad Abdulaziz<sup>†</sup> and Lawrence C. Paulson<sup>‡</sup>

<sup>†</sup>Canberra Research Lab., NICTA, Australia\*; <sup>†</sup>Australian National University; <sup>‡</sup> Computer Laboratory, University of Cambridge, England

**Abstract.** We formalise a statement of Green’s theorem in Isabelle/HOL, which is its first formalisation to our knowledge. The theorem statement that we formalise is enough for most applications, especially in physics and engineering. An interesting aspect of our formalisation is that we neither formalise orientations nor region boundaries explicitly, with respect to the outwards-pointing normal vector. Instead we refer to equivalences between paths.

## 1 Introduction

The *Fundamental Theorem of Calculus* (FTC) is a theorem of immense importance in differential calculus and its applications, relating a function’s differential to its integral. Having been conceived in the seventeenth century in parallel to the development of infinitesimal calculus, more general forms of the FTC have been developed, the most general of which is referred to as the *General Stokes’ Theorem*

A generalisation of the FTC or a special case of the General Stokes’ Theorem in  $\mathbb{R}^2$  was published in 1828 by George Green [2], with applications to electromagnetism in mind. This generalisation is referred to as Green’s Theorem, and it is the main topic of this work. In modern terms the theorem can be stated as follows:

**Theorem 1.** *Given a domain  $D$  with an “appropriate” positively oriented boundary  $\partial D$ , and a field  $F$ , with components  $F_x$  and  $F_y$  “appropriately” defined on  $D$ , the following identity holds:*

$$\int_D \left( \frac{\partial F_y}{\partial x} - \frac{\partial F_x}{\partial y} \right) dx dy = \oint_{\partial D} F_x dx + F_y dy,$$

where the left hand side is a double integral and the right hand side is a line integral<sup>1</sup> in  $\mathbb{R}^2$ .

---

\* NICTA is funded by the Australian Government through the Department of Communications and the Australian Research Council through the ICT Centre of Excellence Program.

<sup>1</sup> This line integral can be physically interpreted as the work done by  $F$  on the  $\partial D$ , making this statement a special case of the 3-dimensional Kelvin-Stokes’ theorem. If the line integral is replaced with  $\oint_{\partial D} F_x dx - F_y dy$ , it can be interpreted as the flux of  $F$  through  $\partial D$  and the theorem would be the 2-dimensional special case of the divergence theorem.

Many statements of Green’s theorem define with varying degrees of generality what is an appropriate boundary (i.e. the geometrical assumptions), and what is an appropriate field (i.e. the analytic assumptions). This mainly depends on how geometrically sophisticated the proof is, and the underlying integral. The prevalent text book form of Green’s theorem asserts that, geometrically, the region can be divided into *elementary regions* and that, analytically, the field is continuous and has continuous partial derivatives throughout the region. Also, usually the underlying integral is a Riemann integral.

Despite this being enough for most applications, more general forms of the theorem have been proved in the analysis literature. Although a full literature review is not appropriate here, we present some examples of very general formulations of Green’s theorem. For example, J.H. Michael [7] proves a statement of the theorem that generalises the geometrical assumptions, where it only assumes that the region has a rectifiable cycle as its boundary. Jurkat et al. [6] prove a statement of the theorem with very general analytic assumptions on the field. They only assume that the field is continuous in the region, and that the *total* derivative of the field exists in the region except for a  $\sigma_1$ -finite set of points in the region. Then, they also derive a very general form of Cauchy’s integral theorem.

Having Green’s theorem formalised is significant because of its wide range of applications, too many to list in full. There are applications in physics (electrodynamics, mechanics, etc.) and engineering (deriving moments of inertia, hydrodynamics, the basis of the planimeter, etc.). In mathematics, Green’s theorem is a fundamental result: it can be used to derive Cauchy’s integral theorem.

We formalise a statement of Green’s theorem for Henstock-Kurtzweil gauge integrals in the interactive theorem prover Isabelle/HOL [8]. Our work builds on the work of Hölzl et.al., where we use the Isabelle/HOL multivariate analysis library [5] and the probability library [4]. We also build on the second author’s porting of John Harrison’s complex analysis library [3]. Our formalisation does not strictly follow a certain proof, but it was inspired by Zorich and Cooke [11], Spivak [10] and Protter [9].

## 2 Basic Concepts and Lemmas

In this section we discuss the basic lemmas we need to prove Green’s theorem. However, we need to firstly discuss two basic definitions needed to state the theorem statement: *line integrals* and *partial derivatives*. Definitions of both of those two concepts are ubiquitous in the literature, nonetheless, we had to adapt them to be defined on the Euclidean spaces type class in the Isabelle multivariate analysis library.

We define the line integral of a function  $F$  on the parameterised path  $\gamma$  as follows:

**Definition 1.** *Line Integral*

$$\int_{\gamma} F \lrcorner_B = \int_0^1 \Sigma_{b \in B} ((F(\gamma(t)) \cdot b)(\gamma'(t) \cdot b)) dt$$

A difference in our definition is that we add the argument  $B$ , a set of vectors, to which  $F$  and  $\gamma$ , and accordingly the line integral are projected. The reasons for adding

the  $B$  argument will become evident later. Above,  $\cdot$  denotes the inner product of two vectors. Formally, the definition is:

```

definition line_integral::
  "('a::euclidean_space  $\Rightarrow$  'a)  $\Rightarrow$  'a set  $\Rightarrow$  (real  $\Rightarrow$  'a)  $\Rightarrow$  real"
where "line_integral F B  $\gamma$  =
  integral {0 .. 1}
  ( $\lambda$ t.  $\sum_{b \in B}. (F(\gamma t) \cdot b) *
  (\text{vector\_derivative } \gamma \text{ (at } t \text{ within } \{0..1\}) \cdot b)$ )"

```

```

definition line_integral_exists where
  "line_integral_exists F B  $\gamma$  =
  (( $\lambda$ t.  $\sum_{b \in B}. F(\gamma t) \cdot b *
  (\text{vector\_derivative } \gamma \text{ (at } t \text{ within } \{0..1\}) \cdot b)$ )
  integrable_on {0..1})"

```

Note that *integral* refers to the Henstock-Kurzweil gauge integral implementation in Isabelle/HOL library. As one would expect, the line integral distributes over unions of sets of vectors and path joins.

```

lemma line_integral_sum_gen:
  assumes "finite B"
  "line_integral_exists F B1  $\gamma$ "
  "line_integral_exists F B2  $\gamma$ "
  "B1  $\cup$  B2 = B" "B1  $\cap$  B2 = {}"
shows "line_integral F B  $\gamma$  =
  (line_integral F B1  $\gamma$ ) + (line_integral F B2  $\gamma$ )"

```

```

lemma line_integral_distrib:
  assumes "line_integral_exists f B  $\gamma$ 1"
  "line_integral_exists f B  $\gamma$ 2"
  "valid_path  $\gamma$ 1" "valid_path  $\gamma$ 2"
shows "line_integral f B ( $\gamma$ 1 +++  $\gamma$ 2) =
  line_integral f B  $\gamma$ 1 + line_integral f B  $\gamma$ 2"

```

The line integral also admits a transformation equivalent to integration by substitution. This lemma applies to paths where all components are defined as a function in terms of one component orthogonal to all of them. It is a critical lemma for proving Green's theorem for "elementary" regions (to be defined later).

```

lemma line_integral_on_pair_path:
  fixes F:: "('a::euclidean_space)  $\Rightarrow$  ('a)"
  g:: "real  $\Rightarrow$  'a"
   $\gamma$ :: "(real  $\Rightarrow$  'a)"
  b:: 'a
  assumes "b  $\in$  Basis"
  " $\forall x. g(x) \cdot b = 0$ "
  " $\gamma = (\lambda t. f(t) *_R b + g(f(t)))$ "
  " $\gamma$  C1_differentiable_on {0 .. 1}"
  "continuous_on (f ` {0..1}) g"
  "(pathstart  $\gamma$ ) \cdot b \leq (\text{pathfinish } \gamma) \cdot b"

```

```

"continuous_on (path_image  $\gamma$ ) ( $\lambda x. F x \cdot b$ ) "
shows "(line_integral F {b}  $\gamma$ )
  = integral (cbox ((pathstart  $\gamma$ )  $\cdot b$ )
              ((pathfinish  $\gamma$ )  $\cdot b$ ))
  ( $\lambda f\_var. (F (f\_var *R b + g(f\_var)) \cdot b)$ )"

```

*Partial derivatives* are defined on the Euclidean space type class implemented in Isabelle/HOL. For a function  $F$  defined on a Euclidean space, we define its partial derivative to be w.r.t. the change in the magnitude of a component vector  $b$  of its input. At a point  $a$ , the partial derivative is defined as:

**Definition 2.** *Partial Derivative*

$$\left. \frac{\partial F(v)}{\partial b} \right|_{v=a} = \left. \frac{dF(a + (x - a \cdot b)b)}{dx} \right|_{x=a \cdot b}$$

Again, this definition is different from the classical definition in that the partial derivative is w.r.t. the change of magnitude of a vector rather than the change in one of the variables on which  $F$  is defined. However, our definition is similar to the classical definition of a partial derivative, when  $b$  is a base vector. Formally we define it as:

```

definition has_partial_vector_derivative::
"('a::euclidean_space  $\Rightarrow$  'b::euclidean_space)  $\Rightarrow$  'a  $\Rightarrow$  'b  $\Rightarrow$  'a
 $\Rightarrow$  bool" where
"has_partial_vector_derivative F b F' a
  = (( $\lambda x. F( (a - ((a \cdot b) *R b)) + x *R b )$ )
     has_vector_derivative F') (at (a  $\cdot$  b))"

```

```

definition partially_vector_differentiable where
"partially_vector_differentiable F b p
  = ( $\exists F'. has\_partial\_vector\_derivative F b F' p$ )"

```

```

definition partial_vector_derivative::
"('a::euclidean_space  $\Rightarrow$  'b::euclidean_space)  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  'b"
where "partial_vector_derivative F b a
  = (vector_derivative
     ( $\lambda x. F( (a - ((a \cdot b) *R b)) + x *R b )$ ) (at (a  $\cdot$  b)))"

```

The following FTC for the partial derivative follows from the FTC for the vector derivative that is proven in Isabelle/HOL analysis library.

```

lemma fundamental_theorem_of_calculus_partial_vector_gen:
fixes k1 k2::"real"
  F::"('a::euclidean_space  $\Rightarrow$  'b::euclidean_space)"
  b::"'a"
  F_b::"('a::euclidean_space  $\Rightarrow$  'b)"
assumes "k1  $\leq$  k2"
  "b  $\cdot$  b = 1"
  "c  $\cdot$  b = 0"
  " $\forall p \in D. has\_partial\_vector\_derivative F b (F_b p) p$ "
  "{v.  $\exists x. k1 \leq x \wedge x \leq k2 \wedge v = x *R b + c$ }  $\subseteq D$ "

```

**shows** "(  $\lambda x. F\_b( x \ *_R \ b + c)$  ) has\_integral  
 $F(k2 \ *_R \ b + c) - F(k1 \ *_R \ b + c)$  ) (cbox k1 k2) "

Given these definitions and basic lemmas, we can now start elaborating on our formalisation of Green's theorem. The first issue is how to formalise  $\mathbb{R}^2$ . We use pairs of *real* to refer to members of  $\mathbb{R}^2$ , where we define the following *locale* that fixes the base vector names:

**locale** *R2* =  
**fixes** *i j*  
**assumes** *i\_is\_x\_axis*: "*i* = (1::real, 0::real)" **and**  
*j\_is\_y\_axis*: "*j* = (0::real, 1::real)"

Proofs of Green's theorem usually start by proving "half" of the theorem statement for every type of "elementary regions" in  $\mathbb{R}^2$ . These regions are referred to as Type I, Type II or Type III regions, defined below.<sup>2</sup>

**Definition 3. Elementary Regions**

A region *D* (modelled as a set of *real* pairs) is Type I iff there are  $C^1$  smooth functions  $g_1$  and  $g_2$  such that for two constants *a* and *b*:

$$D = \{(x, y) \mid a \leq x \leq b \wedge g_2(x) \leq y \leq g_1(x)\}.$$

Similarly *D* would be called type II iff for  $g_1, g_2, a$  and  $b$

$$D = \{(x, y) \mid a \leq y \leq b \wedge g_2(y) \leq x \leq g_1(y)\}.$$

Finally, a region is of type III if it is both of type I and type II.

To prove Green's theorem a common approach is to prove the following two "half" Green's theorems, for any regions  $D_x$  and  $D_y$  that are type I and type II, respectively, and their positively oriented boundaries:

$$\int_{D_x} -\frac{\partial(F_i)}{\partial j} dx dy = \int_{\partial D_x} F \lfloor_{\{i\}},$$

and

$$\int_{D_y} \frac{\partial(F_j)}{\partial i} dx dy = \int_{\partial D_y} F \lfloor_{\{j\}}.$$

Here *i* and *j* are the base vectors while  $F_i$  and  $F_j$  are the *x*-axis and *y*-axis components, respectively, of *F*. However, the difference in the expressions for the type I and type II regions is because of the asymmetry of the *x*-axis and the *y*-axis w.r.t. the orientation. We refer to the top expression as the *x*-axis Green's theorem, and the bottom one as the *y*-axis Green's theorem. Below is the statement of the *x*-axis Green's theorem for type I regions as we have formalised it in Isabelle/HOL. For the boundary, we model its paths explicitly as functions of type *real*  $\Rightarrow$  (*real* \* *real*), where  $\gamma_1, \gamma_2, \gamma_3$  and  $\gamma_4$  are the bottom, right, top and left sides, respectively.

<sup>2</sup> Using elementary regions that are bounded by  $C^1$  smooth functions is as general as using piece-wise smooth functions because it can be shown that the latter can be divided into regions of the former type (see [9]).

```

lemma Greens_thm_type_I:
fixes F::"((real*real)  $\Rightarrow$  (real * real))"
         $\gamma_1 \ \gamma_2 \ \gamma_3 \ \gamma_4$  ::"(real  $\Rightarrow$  (real * real))"
        a::"real" b::"real"
        g1::"(real  $\Rightarrow$  real)" g2::"(real  $\Rightarrow$  real)"
assumes "Dx = {(x,y) . x  $\in$  cbox a b  $\wedge$  y  $\in$  cbox (g2 x) (g1 x)}"
        " $\gamma_1 = (\lambda t. (a + (b - a) * t, g2(a + (b - a) * t)))$ "
        " $\gamma_1$  C1_differentiable_on {0..1}"
        " $\gamma_2 = (\lambda t. (b, g2(b) + t *R (g1(b) - g2(b))))$ "
        " $\gamma_3 = (\lambda t. (a + (b - a) * t, g1(a + (b - a) * t)))$ "
        " $\gamma_3$  C1_differentiable_on {0..1}"
        " $\gamma_4 = (\lambda t. (a, g2(a) + t *R (g1(a) - g2(a))))$ "
        "analytically_valid Dx ( $\lambda p. F(p) \cdot i$ ) j"
        " $\forall x \in$  cbox a b. (g2 x)  $\leq$  (g1 x)"
        "a < b"
shows "(line_integral F {i}  $\gamma_1$ ) + (line_integral F {i}  $\gamma_2$ ) -
        (line_integral F {i}  $\gamma_3$ ) - (line_integral F {i}  $\gamma_4$ ) =
        (integral Dx
        ( $\lambda a. -$  (partial_vector_derivative ( $\lambda p. F(p) \cdot i$ ) j a)))"

```

Proving the lemma above depends on the observation that for a path  $\gamma$  (e.g.  $\gamma_1$  above) that is straight along a vector  $x$  (e.g.  $i$ ),  $\int_{\gamma} F|_{\{x\}} = 0$ , for an  $F$  continuous on  $\gamma$ .

(Formally, this observation follows immediately from theorem *work\_on\_pair\_path*.) The rest of the proof boils down to an application of Fubini's theorem and the FTC to the double integral, the integral by substitution to the line integrals and some algebraic manipulation ([11, page 238]). Nonetheless, this algebraic manipulation proved to be quite tedious when done formally in Isabelle/HOL.

However, we did not discuss the predicate *analytically\_valid*, which represents the analytic assumptions of our statement of Green's theorem, to which an "appropriate" field has to conform. Firstly let  $1_s$  be the indicator function for a set  $s$ . Then, for the  $x$ -axis Green's theorem, our analytic assumptions are that

- (i)  $F_i$  is continuous on  $D_x$
- (ii)  $\frac{\partial(F_i)}{\partial j}$  exists everywhere in  $D_x$
- (iii) the product  $1_{D_x}(x, y) \frac{\partial(F_i)}{\partial j}(x, y)$  is Lebesgue integrable
- (iv) the product  $1_{[a,b]}(x) \int_{g_1(x)}^{g_2(x)} F(x, y) dy$ , where the integral is a Henstock-Kurzweil gauge integral, is a borel measurable function

These assumptions vary symmetrically for the  $y$ -axis Green's theorem, so to avoid having two symmetrical definitions, we defined the predicate *analytically\_valid* to take the axis (as a base vector) as an argument.

```

definition analytically_valid::
  "'a::euclidean_space set  $\Rightarrow$ 
  ('a  $\Rightarrow$  'b::{euclidean_space, times, one, zero})  $\Rightarrow$  'a  $\Rightarrow$  bool"
where
  "analytically_valid s F b =
  (( $\forall a \in s. partially\_vector\_differentiable F b a$ )  $\wedge$ 

```

```

continuous_on s F ∧
integrable lborel
(λp. (partial_vector_derivative F i) p * indicator s p) ∧
(λx. integral UNIV
  (λy. (partial_vector_derivative F i
    (y *R b + x *R (∑ b ∈ (Basis - {i}). b))) *
    (indicator s
      (y *R b + x *R (∑ b ∈ (Basis - {i}). b)) )))
  ∈ borel_measurable lborel) "

```

These conditions refer to Lebesgue integrability and to measurability because we use Fubini's theorem for the Lebesgue integral in Isabelle/HOL's probability library to derive a Fubini like result for the Henstock-Kurzweil integral. Proving Fubini's theorem for the gauge integral would allow for more general analytic assumptions. However, the rest of our approach would still be valid.

We prove the  $y$ -axis Green's theorem for type II regions similarly, where this is its conclusion.

```

shows "(line_integral F {j} γ1) + (line_integral F {j} γ2) -
  (line_integral F {j} γ3) - (line_integral F {j} γ4) =
  (integral Dy
    (partial_vector_derivative (λp. F(p) · j) i))"

```

### 3 More General Structures

Now that we described some of the basic definitions and how to derive Green's theorem for elementary regions, the remaining question is how to prove the theorem for more general regions. As we stated earlier in the introduction, a lot of the text book proofs of Green's theorem are given for regions that can be divided into elementary regions. It can be shown that any *regular* region can be divided into elementary regions [9,11]. Regular regions (see their definition in [9, page 235]), are enough for a lot of applications, especially practical applications in physics and engineering.

In this section we describe how we prove Green's theorem for regions that can be divided both into type I regions and type II regions *only* using vertical and horizontal edges, respectively. We believe that for most practical purposes, the additional assumption that the division is done only by vertical and horizontal edges is equivalent to assuming just the existence of type I and type II divisions. Indeed, we conjecture that the additional constraints do not lead to any loss of generality, however, we would not pursue the proof of this claim in the current paper.

Figure 1 shows an example of a region and its type I and type II partitions. In this example, some of the elementary regions appear to have a missing edge. This is because the type I or the type II partitioning induced a *one-point path*: a function mapping the interval  $[0, 1]$  to a single point in  $\mathbb{R}^2$ . For instance, the left edge in the left 1-chain in 1b is a point on the  $x$ -axis.




Fig. 1: An annulus and its partitioning in type I and type II regions. In this figure every 1-cube (i.e. path) is represented with an arrow whose direction is the same as the orientation of the 1-cube. a) The positively oriented boundary of the annulus. b), c) and d) The members of a type I partitioning of the annulus. e) A 1-chain that includes all the horizontal boundaries in the type I partition. f), g) and h) The members of a type II partitioning of the annulus. i) A 1-chain that includes all the vertical boundaries in the type II partition. j) A common subdivision of the chains in e) and i).

### 3.1 Chains and cubes

For tackling more general regions and their boundaries we use the concepts of cubes and chains [10, chapter 8]. One use of cubes is to represent parameterisable (sub)surfaces (regions in  $\mathbb{R}^2$  and paths in our case). A  $k$ -dimensional such surface embedded in  $\mathbb{R}^n$  is represented by a function whose domain is a space homeomorphic to  $\mathbb{R}^k$  and whose codomain is  $\mathbb{R}^n$ . Roughly speaking, we model cubes as functions and chains as sets of cubes. We use the existing Isabelle/HOL formalisation of paths, where we model 1-cubes as functions defined on the interval  $\{0..1\}$ . We model a 1-chain as a set of pairs of *int* (coefficients) and 1-cubes. For example, the following definition shows the lifting of the line integrals to 1-chains.

```

definition one_chain_line_integral ::
  "(real*real) => (real *real) => ((real*real) set) =>
    (int * (real =>real*real)) set => real" where
  "one_chain_line_integral F b C =
    ( $\sum_{(k,\gamma)\in C. k * (line\_integral\ F\ b\ \gamma)}$ )"

```

We extend the way we model 1-cubes to model 2-cubes, which we model as functions of type  $(real * real \Rightarrow real * real)$  defined on the interval  $\{(0,0)..(1,1)\}$ .

```

definition cubeImage ::
  "(real*real => real*real) => ((real*real) set)" where
  "cubeImage twoC = (twoC ` (cbox (0,0) (1,1)))"

```

The orientation of the boundary of a 2-cube (a 1-chain) is taken to be counter-clockwise. A 1-cube is given the coefficient  $-1$  if the path's direction is against the counter-clockwise boundary traversal order, otherwise it is given the coefficient  $1$ . Formally this is defined as follows:

```

definition horizontal_boundary ::
  "(real*real => real*real) => (real* (real => real*real)) set"
where "horizontal_boundary twoC =
  {(1, ( $\lambda x. twoC(x,0)$ )), (-1, ( $\lambda x. twoC(x,1)$ ))}"

```

```

definition vertical_boundary ::
  "(real*real => real*real) => (real* (real => real*real)) set"
where "vertical_boundary twoC =
  {(-1, ( $\lambda y. twoC(0,y)$ )), (1, ( $\lambda y. twoC(1,y)$ ))}"

```

```

definition boundary where
  "boundary twoC =
    horizontal_boundary twoC  $\cup$  vertical_boundary twoC"

```

We follow the convention and define the 2-cubes in such a way that the top and left edges are against the counter-clockwise orientation (e.g. see the 2-cube in Figure 1c). Accordingly both the left and top edges take a  $-1$  coefficient in the 1-cube representation. Defining 2-cubes in that way makes it easier to define predicates identifying type I and type II 2-cubes, as follows.

```

definition typeI_twoCube ::
  "((real * real)  $\Rightarrow$  (real * real))  $\Rightarrow$  bool" where
  "typeI_twoCube twoC  $\longleftrightarrow$ 
    ( $\exists$  (a::real) (b::real) (g1::real $\Rightarrow$ real) (g2::real  $\Rightarrow$  real).
      a < b  $\wedge$  ( $\forall$ x  $\in$  {a..b}. g2 x  $\leq$  g1 x)  $\wedge$ 
      twoC =
        ( $\lambda$ (t1,t2). ((1-t1)*a + t1*b,
                    (1 - t2) * (g2 ((1-t1)*a + x*b)) +
                    t2 * (g1 ((1-t1)*a + t1*b))))  $\wedge$ 
      g1 C1_differentiable_on {a .. b}  $\wedge$ 
      g2 C1_differentiable_on {a .. b})"

```

Although we do not render it here, an analogous predicate is defined for type II 2-cubes. We also require that all 2-cubes conform to the following predicate:

```

definition valid_two_cube where
  "valid_two_cube twoC = (card (boundary twoC) = 4)"

```

This predicate filters out cases where 2-cubes have either: i) right and top edges that are both one point paths, or ii) left and bottom edges that are both one point paths. Although this assumption potentially leaves our theorems less general regarding some corner cases, it makes our computations much smoother. After defining these concepts on 2-cubes, we derive the following statement of  $x$ -axis Green's theorem (and its equivalent  $y$ -axis Green's theorem) in terms of 2-cubes.

```

GreenThm_typeI_twoCube:
  assumes "typeI_twoCube twoC"
  "valid_two_cube twoC"
  "analytically_valid (cubeImage twoC) ( $\lambda$ x. (F x)  $\cdot$  i) j"
  shows "integral (cubeImage twoC)
    ( $\lambda$ a. - partial_vector_derivative ( $\lambda$ p. F p  $\cdot$  i) j a)
    = one_chain_line_integral F {i} (boundary twoC)"

```

Although we anticipated that proving this theorem would be a straightforward unfolding of definitions and usage of `GreenThm_typeI`, it was a surprisingly long and tedious proof that took a few hundred lines.

For 2-chains, we model them as sets of 2-cubes, which suits our needs in working in the context of Green's theorem. We define the boundary of a 2-chain as follows:

```

definition two_chain_boundary::
  "(((real*real)  $\Rightarrow$  (real*real))) set)  $\Rightarrow$ 
  ((real* (real  $\Rightarrow$  (real*real))) set)" where
  "two_chain_boundary twoChain =  $\bigcup$  (boundary ` twoChain)"

```

We similarly defined the functions `two_chain_horizontal_boundary` and `two_chain_vertical_boundary`. We also lift the double integral to 2-chains as follows.

```

definition two_chain_integral::
  "(((real*real)  $\Rightarrow$  (real*real))) set)  $\Rightarrow$ 
  ((real*real) $\Rightarrow$ (real))  $\Rightarrow$  real" where
  "two_chain_integral twoChain F =
    ( $\sum$  (C) $\in$ twoChain. (integral (cubeImage C) F))"

```

Lastly, to smoothe our computations on integrals over 2-chains and their boundaries, we require that a 2-chain: i) only has valid 2-cubes members, ii) edges of different 2-cubes only coincide if they have opposite orientations, and iii) different 2-cubes have different images. These requirements are formally defined in the following predicate:

**definition** *valid\_two\_chain* **where**

```
"valid_two_chain twoChain =
  (∀twoCube ∈ twoChain. valid_two_cube twoCube) ∧
  (∀twoCube1 ∈ twoChain.
    ∀twoCube2 ∈ twoChain.
      twoCube1 ≠ twoCube2 →
        boundary twoCube1 ∩ boundary twoCube2 = {}) ∧
  inj_on cubeImage twoChain)"
```

Given these definitions on 2-chains, we lift our  $x$ -axis Green's theorem from 2-cubes to 2-chains, as shown in the following statement.

**lemma** *GreenThm\_typeI\_twoChain*:

**assumes** "∀twoCube ∈ twoChain. typeI\_twoCube twoCube"

"valid\_two\_chain twoChain"

"∀twoC ∈ twoChain.

analytically\_valid (cubeImage twoC) (λx. (F x) · i) j"

**shows** "two\_chain\_integral twoChain

(λa. - partial\_vector\_derivative  
(λx. (F x) · i) j a) =

one\_chain\_line\_integral F {i}

(two\_chain\_boundary twoChain)"

A similar lemma can be derived for a 2-chain whose members are all of Type II (with the obvious consequences of orientation asymmetry).

After proving the  $x$ -axis and  $y$ -axis Green's theorems, the next step is, after algebraic and analytic manipulation, to add the line integral sides of the  $x$ -axis Green's theorem to its counterpart in the  $y$ -axis theorem and similarly add the double integrals of both theorems. Given *GreenThm\_typeI\_twoChain* and its type II equivalent, we can sum up both sides of the equalities in the conclusion and get Green's theorem in terms of 2-chains and their boundaries. However, the main goal of the paper is to obtain the theorem directly for a region and its boundary, given that the region can be *vertically* sliced into regions of type I and *horizontally* sliced into regions of type II.

The first (and easier) part in proving this is to prove the equivalence of the double integral on a region and the integral on a 2-chain that divides that region. Before deriving such a theorem we generalised the notion of *division\_of*, defined in Isabelle/HOL's multivariate analysis library, to work when the division is not constituted of rectangles.

**definition** *gen\_division* **where**

```
"gen_division s S =
  (finite S ∧ (⋃ S = s) ∧
  (∀u ∈ S. ∀t ∈ S. u ≠ t → negligible (u ∩ t)))"
```

Then we show the following equivalence:

```

lemma two_chain_integral_eq_integral_divisible:
assumes "∀ twoCube ∈ twoChain. F integrable_on cubeImage twoCube"
          "gen_division s (cubeImage ` twoChain)"
          "valid_two_chain twoChain"
shows "integral s F = two_chain_integral twoChain F"

```

The other part concerning the line integrals, proved to be trickier, and we will explain it in the next section.

### 3.2 Dealing with Boundaries

What remains now is to prove an equivalence between the line integral on the 1-chain boundary of the region under consideration and the line integral on the 1-chain boundary of the region's elementary division (i.e. the 2-chain division of the region). The classical approach depends on that the line integrals on the introduced boundaries will cancel each other, leaving out the line integral on the region's original boundary. For example, the vertical-straight-line paths in Figures 1b, 1c and 1d, are the introduced boundaries to obtain the type I division of the annulus. In this example, the line integrals on the introduced vertical-straight-line paths will cancel each other because of their opposite orientations, thus, leaving out the integral on the original boundary.

To prove this formally, the classical approach needs to define what is a positively oriented boundary, which requires an explicit definition of the boundary of a region, and also defining the exterior normal of the region (examples of this approach can be found in [9,1]). However, we use a different approach that does not depend on these definitions and avoids a lot of the resulting geometrical and analytic complications. Our approach depends on two observations:

- O1: if a path  $\gamma$  is straight along a vector  $x$ , then  $\int_{\gamma} F|_{\{x\}} = 0$ , for an  $F$  continuous on  $\gamma$ .
- O2: partitioning the region in type I/type II regions was done by introducing only vertical/horizontal boundaries.

For a type I 2-chain division of a region, consider the 1-chain  $\gamma_x$ , that: i) includes *all* the horizontal boundaries of the dividing 2-chain, and ii) includes *some* subpaths of the vertical boundaries of the dividing 2-chain (call this condition Cx). Based on O1, the line integral on the vertical edges in the 1-chain boundary of the type I division and accordingly  $\gamma_x$ , projected on  $i$ , will be zero. Accordingly we can prove the  $x$ -axis Green's theorem for  $\gamma_x$ . Formal statements of Cx (formally: *only\_vertical\_division*), and the consequence of a 1-chain conforming to it are:

```

definition only_vertical_division where
"only_vertical_division one_chain two_chain =
(∃ V. finite V ∧
  (∀ (k, γ) ∈ V.
    (∃ (k', γ') ∈ two_chain_vertical_boundary two_chain.
      (∃ a ∈ {0 .. 1}.
        ∃ b ∈ {0..1}.
          a ≤ b ∧ subpath a b γ' = γ))) ∧
one_chain = V ∪ (two_chain_horizontal_boundary two_chain))"

```

```

lemma GreenThm_typeI_divisible_region_boundary_gen:
assumes "valid_typeI_division s twoChain"
         "∀twoC ∈ twoChain.
          analytically_valid (cubeImage twoC) (λa. F(a) · i) j"
         "only_vertical_division γ_x twoChain"
shows "integral s
        (λx. - partial_vector_derivative (λa. F(a) · i) j x)
        = one_chain_line_integral F {i} γ_x "

```

In the previous lemma, *valid\_typeI\_division* is an abbreviation that, for a region and a 2-chain, means that the 2-chain constitutes only valid type I cubes and that this 2-chain is a division of the given region.

An analogous condition, *Cy*, for a type II partitioning asserts that the 1-chain includes *all* the vertical boundaries of the dividing 2-chain, and includes *some* subpaths of the horizontal boundaries of the dividing 2-chain. For *Cy*, we formally define the predicate (*only\_horizontal\_division*) and prove an analogous theorem for type II partitions, with the obvious changes in the conclusion.

From the second observation, *O2*, we can conclude that there will always be

- a 1-chain,  $\gamma_x$ , whose image is the boundary of the region under consideration and that satisfies *Cx* for the type I division.
- a 1-chain,  $\gamma_y$ , whose image is the boundary of the region under consideration and that satisfies *Cy* for the type II division, where it is *not* necessary that  $\gamma_x = \gamma_y$ .

Figure 1e and Figure 1i show two 1-chains that satisfy *Cx* and *Cy* for the type I and type II divisions of the annulus. Notice that in this example, those two 1-chains are *not equal* even though they have the same *orientation* and *image*.

Now, if we can state and formalise the equivalence between  $\gamma_x$  and  $\gamma_y$ , and that this equivalence lifts to equal line integrals, we can obtain Green’s theorem in terms of the region, which is our goal. One way to formalise path equivalence is to explicitly define the notion of orientation. Then the equivalence between  $\gamma_x$  and  $\gamma_y$  can be characterised by their having similar orientations and images. An advantage of this approach is that it can capture equivalence in path orientations regardless of the path image.

However, we do not need this generality in the context of proving the equivalence of 1-chains that have the same image and orientation, especially that this generality will cost a lot of analytic and geometric complexities to be formalised. Instead we choose to formalise the notion of equivalence in terms of having a *common subdivision*. For example the 1-chain shown in Figure 1j is a subdivision of each of the 1-chains in Figure 1e and Figure 1i as well as the original boundary 1-chain in Figure 1a, i.e. a common subdivision between the three 1-chains. We now formally define the concept of a common subdivision between 1-chains, where we mainly focus on “boundary” 1-chains, defined as follows.

```

definition boundary_chain where
"boundary_chain s = (∀ (k, γ) ∈ s. k = 1 ∨ k = -1) "

```

First, we lift the *path\_join* operator defined in the Isabelle/HOL multivariate analysis library, to act on 1-chains ordered into lists as follows.

```

fun rec_join ::
  "(int *(real  $\Rightarrow$  (real*real))) list  $\Rightarrow$  (real  $\Rightarrow$  (real * real))"
where
  "rec_join [] = ( $\lambda$ x.0)" |
  "rec_join ((k,  $\gamma$ ) # xs) =
    (if (k = 1) then  $\gamma$  else reversepath  $\gamma$ )  +++ (rec_join xs)"

```

To use the theory for paths developed in the multivariate analysis library, we need the joined chains to be piece-wise  $C^1$  smooth in the sense that is defined in that library (*valid\_path*). A necessary condition for a path to be valid, is that the ending point of every piece of that path to be the starting point of the next. Accordingly we define the following predicate for the validity 1-chains ordered into lists.

```

fun valid_chain_list where
  "valid_chain_list [] = True" |
  "valid_chain_list ((k,  $\gamma$ )#l) =
    (( if (k = 1) then (pathfinish  $\gamma$  = pathstart (rec_join l))
      else (pathfinish (reversepath  $\gamma$ ) = pathstart (rec_join l)))
     $\wedge$  valid_chain_list l)"

```

Based on those concepts we now define what it means for a 1-chain to be a subdivision of a path, which is a straightforward definition.

```

definition chain_subdiv_path::
  "(real  $\Rightarrow$  real * real)  $\Rightarrow$  ((int * (real  $\Rightarrow$  real * real)) set)
   $\Rightarrow$  bool" where
  "chain_subdiv_path  $\gamma$  one_chain
    = ( $\exists$ l. set l = one_chain  $\wedge$  distinct l  $\wedge$  rec_join l =  $\gamma$   $\wedge$ 
      valid_chain_list l)"

```

We call a 1-chain  $\gamma$ , a subdivision of another 1-chain  $\eta$ , if one can map every cube in  $\eta$  to a sub-chain of  $\gamma$  that is a subdivision of it. Formally this is defined as follows:

```

definition chain_subdiv_chain where
  "chain_subdiv_chain one_chain1 subdiv =
    ( $\exists$ f. (( $\bigcup$  (f ` one_chain1)) = subdiv)  $\wedge$ 
      ( $\forall$  (k, $\gamma$ ) $\in$ one_chain1.
        if k = 1 then
          chain_subdiv_path  $\gamma$  (f(k,  $\gamma$ ))
        else
          chain_subdiv_path (reversepath  $\gamma$ ) (f(k, $\gamma$ )))  $\wedge$ 
      ( $\forall$ p $\in$ one_chain1.
         $\forall$ p' $\in$ one_chain1. p  $\neq$  p'  $\longrightarrow$  f p  $\cap$  f p' = {})  $\wedge$ 
      ( $\forall$ x  $\in$  one_chain1. finite (f x)))"

```

After proving that each of the previous notions of equivalence implies equality of line integrals, we define equivalence of 1-chains in terms of having a common subdivision, and prove that it implies equal line integrals. We define it as having a boundary 1-chain that is a subdivision for each of the 1-chains under consideration. Formally this definition and the equality of line integrals that it implies are as follows:

**definition** *common\_boundary\_sudivision\_exists* **where**  
`"common_boundary_sudivision_exists one_chain1 one_chain2 =`  
`( $\exists$  subdiv. chain_sudiv_chain one_chain1 subdiv  $\wedge$`   
`chain_sudiv_chain one_chain2 subdiv  $\wedge$`   
`( $\forall$  (k,  $\gamma$ )  $\in$  subdiv. valid_path  $\gamma$ )  $\wedge$`   
`(boundary_chain subdiv))"`

**lemma** *common\_sudivision\_imp\_eq\_line\_integral*:  
**assumes** `"(common_boundary_sudivision_exists`  
`one_chain1 one_chain2)"`  
`"boundary_chain one_chain1"`  
`"boundary_chain one_chain2"`  
`" $\forall$  (k,  $\gamma$ )  $\in$  one_chain1. line_integral_exists F B  $\gamma$ "`  
`"finite one_chain1"`  
`"finite one_chain2"`  
**shows** `"one_chain_line_integral F B one_chain1 =`  
`one_chain_line_integral F B one_chain2"`  
`" $\forall$  (k,  $\gamma$ )  $\in$  one_chain2. line_integral_exists F B  $\gamma$ "`

Based on this lemma, finally, we prove the following statement of Green's theorem.

**lemma** *GreenThm\_typeI\_typeII\_divisible\_region*:  
**assumes** `"valid_typeI_division s twoChain_typeI"`  
`"valid_typeII_division s twoChain_typeII"`  
`" $\forall$  twoC  $\in$  twoChain_typeI.`  
`analytically_valid (cubeImage twoC) ( $\lambda$ x. (F x)  $\cdot$  i) j"`  
`" $\forall$  twoC  $\in$  twoChain_typeII.`  
`analytically_valid (cubeImage twoC) ( $\lambda$ x. (F x)  $\cdot$  j) i"`  
`"only_vertical_division  $\gamma_x$  twoChain_typeI"`  
`"boundary_chain  $\gamma_x$ "`  
`"only_horizontal_division  $\gamma_y$  twoChain_typeII"`  
`"boundary_chain  $\gamma_y$ "`  
`"common_boundary_sudivision_exists  $\gamma_x$   $\gamma_y$ "`  
**shows** `"integral s`  
`( $\lambda$ x. partial_vector_derivative ( $\lambda$ x. (F x)  $\cdot$  j) i x -`  
`partial_vector_derivative ( $\lambda$ x. (F x)  $\cdot$  i) j x)`  
`= one_chain_line_integral F {i, j}  $\gamma_x$ "`  
`"integral s`  
`( $\lambda$ x. partial_vector_derivative ( $\lambda$ x. (F x)  $\cdot$  j) i x -`  
`partial_vector_derivative ( $\lambda$ x. (F x)  $\cdot$  i) j x)`  
`= one_chain_line_integral F {i, j}  $\gamma_y$ "`

This theorem does not require the 1-chains  $\gamma_x$  and  $\gamma_y$  to have as their image exactly the boundary of the region. However, of course it applies to the 1-chains if their image is the boundary of the region. Accordingly it fits as Green's theorem for a region that can be divided into elementary regions just by vertical and horizontal slicing.

It is worth noting that although this statement seems to have a lot of assumptions, its analytic assumptions regarding the field are strictly more general than the those in [11,9], where they require the field and both of its partial derivatives to be continuous in the region. For the geometric assumptions, on the other hand, we have two extra

requirements: the type I and type II divisions should be obtained using only vertical slicing and only horizontal slicing, respectively.

## 4 Conclusion and Future Work

We formalised a statement of Green's theorem that is enough for a lot of practical applications. Theory and concepts we developed here can be used in proving more general statements of Green's theorem [6,7]. Most such proofs depend on approximating both line and double integrals on a region by corresponding integrals on a region that can be divided into elementary regions. An interesting aspect of our work is that we avoided defining the region's boundary and its orientation explicitly. We did so by assuming that the division was done by inserting only vertical edges for the type I division, and only horizontal edges for the type II division. We claim that this added condition on the division represents no loss of generality, and intend to prove this claim in the future.

*Isabelle Notation and Availability* All blocks starting with isabelle keywords: **lemma**, **definition**, **fun** have been generated automatically using Isabelle/HOL's  $\LaTeX$  pretty-printing utility. Sometimes we have edited them slightly to improve readability, but the full sources are available online.<sup>3</sup>

*Acknowledgement* This research was supported in part by an Australian National University - International Alliance of Research Universities Travel Grant and by an Australian National University, College of Engineering and Computer Science Dean's Travel Grant Award. Also, the first author thanks Katlyn Quenzer for helpful discussions.

## References

1. Federer, H.: Geometric measure theory. Springer (2014)
2. Green, G.: An essay on the application of mathematical analysis to the theories of electricity and magnetism. (1828)
3. Harrison, J.: Formalizing basic complex analysis. From Insight to Proof: Festschrift in Honour of Andrzej Trybulec. Studies in Logic, Grammar and Rhetoric 10(23), 151–165 (2007)
4. Hölzl, J., Heller, A.: Three chapters of measure theory in Isabelle/HOL. In: Interactive Theorem Proving, pp. 135–151. Springer (2011)
5. Hölzl, J., Immler, F., Huffman, B.: Type classes and filters for mathematical analysis in Isabelle/HOL. In: Interactive Theorem Proving, pp. 279–294. Springer (2013)
6. Jurkat, W., Nonnenmacher, D.: The general form of Green's theorem. Proceedings of the American Mathematical Society 109(4), 1003–1009 (1990)
7. Michael, J.: An approximation to a rectifiable plane curve. Journal of the London Mathematical Society 1(1), 1–11 (1955)
8. Nipkow, T., Paulson, L.C., Wenzel, M.: Isabelle/HOL: a proof assistant for higher-order logic, vol. 2283. Springer (2002)
9. Protter, M.H.: Basic elements of real analysis. Springer Science & Business Media (2006)
10. Spivak, M.: A Comprehensive Introduction to Differential Geometry. Publish or Perish, Inc., University of Tokyo Press (1981)
11. Zorich, V.A., Cooke, R.: Mathematical analysis II. Springer (2004)

---

<sup>3</sup> [bitbucket.org/MohammadAbdulaziz/isabellegeometry/](http://bitbucket.org/MohammadAbdulaziz/isabellegeometry/)